Computer Algebra for Physics Examples

Electrostatics, Magnetism, Circuits and Mechanics of Charged Particles

Part 4

Mechanics of Charged Particles

Leon Magiera and Josef Böhm

V. Mechanics of Charged Particles

The Mechanics of a Charge in Electric, Magnetic and Quasi-stationary Electromagnetic Field

Introduction

In the majority of problems related to the mechanics of electrically charged particles in quasi-stationary electromagnetic fields it is sufficient to take into account the action of the forces

$$\vec{F} = q\,\vec{E} + q(\vec{v} \times \vec{B}),$$

where: q - the charge of a particle,
$\vec{E}$ - the vector of the electrical field strength,
$\vec{v}$ - the velocity of the particle,
$\vec{B}$ - the vector of the magnetic field.

The first term results form the interaction of the charge with the electric field, the second one – the *Lorentz* force – acts upon a charge moving within a magnetic field. Charge and mass are inseparable. However, we will only deal with problems with negligible gravitational force in relation to the electromagnetic force.

In the examples considered, it will be assumed that the velocity of a particle is not large. Thus, we can ignore relativistic effects.

All files are available on request.

nojo.boehm@pgv.at

leon.magiera@wp.pl

V.1 Find the path of a particle of mass *m* and charge *q* moving in a constant and uniform electric field of strength $\vec{E}$.

The initial conditions are: position $\vec{r}_0 = (0,0,0)$ and velocity $\vec{v} = (v_{0x}, v_{0y}, 0)$.

Solution:

The electric force acting on a particle is given by

$$\vec{F} = q\,\vec{E}.$$

According *Newton*'s equation we obtain

$$m\frac{d^2\vec{r}}{dt^2} \equiv m\ddot{\vec{r}} = q\,\vec{E},$$

where $\vec{r}$ is the position vector of mass *m*.

We enter the position vector and the field vector followed by the definition of the system of coordinates as given in Fig. V.1 and the velocity vector.



Fig. V.1

```
(%i2)   r_(t):=[x(t),y(t),z(t)]$   E_:[0,E,0]$
(%i7)   x0:0$   y0:0$   z0:0$
        v0x:v0*cos(α)$   v0y:v0*sin(α),  v0z:0$
```

Newton's equation in vector notation has the form

```
(%i8)   de_:diff(r_(t),t,2)=q/m*E_;
```

$$\text{(de\_)}\quad \left[\frac{d^2}{d\,t^2}\,x(t),\ \frac{d^2}{d\,t^2}\,y(t),\ \frac{d^2}{d\,t^2}\,z(t)\right]=\left[0,\ \frac{q\,E}{m},\ 0\right]$$

Let's extract all its components:

```
(%i11)  dex2:lhs(de_)[1]=rhs(de_)[1];
        dey2:lhs(de_)[2]=rhs(de_)[2];
        dez2:lhs(de_)[3]=rhs(de_)[3];
```

$$\text{(dex2)}\quad \frac{d^2}{d\,t^2}\,x(t)=0$$

$$\text{(dey2)}\quad \frac{d^2}{d\,t^2}\,y(t)=\frac{q\,E}{m}$$

$$\text{(dez2)}\quad \frac{d^2}{d\,t^2}\,z(t)=0$$

The above simple differential equations can be solved "by hand", however we apply computer algebra. As the solutions of equations `dex2` and `dez2` are trivial, we present only solution methods for equation `dey2`.

Method 1:

Using integration twice.

```
Method 1
(%i12)  integrate(dey2,t);
```

$$(\%o12)\quad \frac{d}{d\,t}\,y(t)=\frac{q\,t\,E}{m}+\%c1$$

It can easily be seen that constant %c1 means the initial velocity component *v0y*. We perform the appropriate substitution and integrate the second time.

```
(%i13)  subst(%c1=v0y,%);
```

$$(\%o13)\quad \frac{d}{d\,t}\,y(t)=v0\,\sin(\alpha)+\frac{q\,t\,E}{m}$$

```
(%i14)  integrate(%,t);
```

$$(\%o14)\quad y(t)=t\,v0\,\sin(\alpha)+\frac{q\,t^2\,E}{2\,m}+\%c2$$

As the *y*-component of the initial position vector is zero we substitute again and receive *y*(*t*).

```
(%i15)  subst(%c2=0,%);
```

$$(\%o15)\quad y(t)=t\,v0\,\sin(\alpha)+\frac{q\,t^2\,E}{2\,m}$$

Method 2:

Applying `ode2` to get the general solution first followed by `ic2` to obtain the particular solution.

```
Method 2
(%i16)  ode2(dey2,y(t),t);
```

$$(\%o16)\quad y(t)=\frac{q\,t^2\,E}{2\,m}+\%k2\,t+\%k1$$

```
(%i17)  ic2(%,t=0,y(t)=0,diff(y(t),t)=v0y);
```

$$(\%o17)\quad y(t)=t\,v0\,\sin(\alpha)+\frac{q\,t^2\,E}{2\,m}+y(0)$$

Method 3:

Applying `desolve` to find the general solution for all DEs followed by `atvalue` to obtain the particular solutions.

```
Method 3
(%i18) gsoln:desolve([dex2,dey2,dez2],[x(t),y(t),z(t)])$
(%i19) psoln:subst([x(0)=x0,y(0)=y0,z(0)=z0,
                    at('diff(x(t),t),t=0)=v0*cos(α),
                    at('diff(y(t),t),t=0)=v0*sin(α),
                    at('diff(z(t),t),t=0)=0],gsoln);
```

(psoln) $[x(t)=t\, v0\cos(\alpha), y(t)=t\, v0\sin(\alpha)+\frac{q\, t^2 E}{2\, m}, z(t)=0]$

The obtained result indicates that the motion of the charged particle lies within the *xy*-plane. In order to get the relation describing the trajectory of the particle in explicit form we try to eliminate the time parameter *t* from the above equations for $x(t)$ and $y(t)$.

```
(%i20) solve(eliminate([x=rhs(psoln[1]),y=rhs(psoln[2])],[t]),y);
```

(%o20) $[y=\frac{2\, m\, v0^2\, x\cos(\alpha)\sin(\alpha)+q\, x^2 E}{2\, m\, v0^2\cos(\alpha)^2}]$

We rewrite the result as function depending on x, q and α.

```
(%i21) y(y,q,α):=(2*m*v0^2*x*cos(α)*sin(α)+q*x^2*E)/
                  (2*m*v0^2*cos(α)^2);
```

(%o21) $y(y,q,\alpha):=\frac{2\, m\, v0^2\, x\cos(\alpha)\sin(\alpha)+q\, x^2 E}{2\, m\, v0^2\cos(\alpha)^2}$

Let's enter a set of parameter values for plotting some trajectories:

```
(%i24) E:10$  v0:2$  m:3$
(%i25) y(x,q,α);
```

(%o25) $\frac{24\, x\cos(\alpha)\sin(\alpha)+10\, q\, x^2}{24\cos(\alpha)^2}$

We recognize the trajectories as parabolas and plot for $q=\pm 1$ and $\alpha=-\frac{\pi}{4},0,\frac{\pi}{4}$ (which is the angle of the original direction of motion with the positive *x*-axis).

```
(%i32) y1:y(x,1,-%pi/4)$ y2:y(x,1,0)$ y3:y(x,1,%pi/4)$
       y4:y(x,-1,-%pi/4)$ y5:y(x,-1,0)$ y6:y(x,-1,%pi/4)$
       plot2d([y1,y2,y3,y4,y5,y6],[x,0,2],[y,-4,4],
           [legend,"q=1,alpha=-PI/4","q=1,alpha=0","q=1,alpha=PI/4",
              "q=-1,alpha=-PI/4","q=-1,alpha=0","q=-1,alpha=PI/4"])$
plot2d: some values were clipped.
 plot2d: some values were clipped.
```

You can find the graphs of the trajectories on the next page.



It is no problem to perform methods 1 and 2 working with DERIVE and TI-NspireCAS as well. Both systems don't provide an appropriate tool like presented in %i18 and %i19 above. *MATHEMATICA* does.

See below a possible TI-NspireCAS treatment inspired by method 3.

$r\_(t):=[x(t)\ \ y(t)\ \ z(t)]$	*Done*
$ee\_:=[0\ \ ee\ \ 0]$	$[0\ \ ee\ \ 0]$
$x0:=0:y0:=0:z0:=0$	0
$v0x:=v0\cdot\cos(\alpha):v0y:=v0\cdot\sin(\alpha):v0z:=0$	0
⚠ $de\_:=\frac{d^2}{dt^2}(r\_(t))=\frac{q}{m}\cdot ee\_$	$\left[\frac{d^2}{dt^2}(x(t))=0\ \ \frac{d^2}{dt^2}(y(t))=\frac{q\cdot ee}{m}\ \ \frac{d^2}{dt^2}(z(t))=0\right]$
$\text{deSolve}(x''=0 \text{ and } x(0)=x0 \text{ and } x'(0)=v0x,t,x)$	$x=v0\cdot\cos(\alpha)\cdot t$
$\text{deSolve}\left(y''=\frac{q\cdot ee}{m} \text{ and } y(0)=0 \text{ and } y'(0)=v0\cdot\sin(\alpha),t,y\right)$	$y=\frac{q\cdot ee\cdot t^2}{2\cdot m}+v0\cdot\sin(\alpha)\cdot t$
$\text{deSolve}(z''=0 \text{ and } z(0)=z0 \text{ and } z'(0)=v0z,t,z)$	$z=0$
$\text{solve}\left(x=v0\cdot\cos(\alpha)\cdot t \text{ and } y=\frac{q\cdot ee\cdot t^2}{2\cdot m}+v0\cdot\sin(\alpha)\cdot t,\{y,t\}\right)$	
	$t=\frac{x}{v0\cdot\cos(\alpha)} \text{ and } y=\frac{x\cdot(q\cdot ee\cdot x+2\cdot m\cdot v0^2\cdot\sin(\alpha)\cdot\cos(\alpha))}{2\cdot m\cdot v0^2\cdot(\cos(\alpha))^2} \text{ and } m\neq 0$

It's a question of taste what to prefer: the "two command"-procedure of Maxima – which might be a bit complicated for one or the other "CASer", or the slower – but clearer from my point of view – "step-for-step"-procedure of TI-NspireCAS. You can see how to find the explicit form of the trajectories.

V.2 An electron of mass m and charge q is moving in a constant magnetic field. The initial conditions are: position $\vec{r}_0 = (0,0,0)$ and velocity $\vec{v} = (v_{0x}, v_{0y}, 0)$.

a) Calculate and analyze the trajectory of the electron.

b) Calculate the *charge-to-mass ratio q/m* of the electron.

Solution:

We refer to the formula given in the introduction

$$\vec{F} = q\vec{E} + q(\vec{v} \times \vec{B}),$$

In our problem is $|\vec{E}| = 0$, so we have

$$\vec{F} = q(\vec{v} \times \vec{B}) \text{ with } \vec{B}(B,0,0).$$

The Maxima computation procedure follows

```
(%i1)   r_(t):=[x(t),y(t),z(t)]$
(%i2)   [vx(t):='diff(x(t),t),vy(t):='diff(y(t),t),
         vz(t):='diff(z(t),t)];
```

$$\text{(\%o2)}\quad [vx(t) := \frac{d}{dt}x(t), vy(t) := \frac{d}{dt}y(t), vz(t) := \frac{d}{dt}z(t)]$$

```
(%i4)   v_(t):=[vx(t),vy(t),vz(t)]$
        B_:[B,0,0]$
```

As shown in part 1 we have to load the library "vect" in order to have the vector product available which must be entered by a "~" (%i16).

```
(%i6)   load(vect)$
        de_:diff(r_(t),t,2)=q/m*express(v_(t)~B_);
```

$$\text{(de\_)}\quad [\frac{d^2}{dt^2}x(t), \frac{d^2}{dt^2}y(t), \frac{d^2}{dt^2}z(t)] = [0, \frac{q\left(\frac{d}{dt}z(t)\right)B}{m}, -\frac{q\left(\frac{d}{dt}y(t)\right)B}{m}]$$

We proceed in the same way as in problem V.1, starting with extracting the components of *Newton*'s equation:

```
(%i9)   dex2:lhs(de_)[1]=rhs(de_)[1];
        dey2:lhs(de_)[2]=rhs(de_)[2];
        dez2:lhs(de_)[3]=rhs(de_)[3];
```

$$\text{(dex2)}\quad \frac{d^2}{dt^2}x(t) = 0$$

$$\text{(dey2)}\quad \frac{d^2}{dt^2}y(t) = \frac{q\left(\frac{d}{dt}z(t)\right)B}{m}$$

$$\text{(dez2)}\quad \frac{d^2}{dt^2}z(t) = -\frac{q\left(\frac{d}{dt}y(t)\right)B}{m}$$

First we apply `desolve` for calculating the general solution. Then we find the particular solution and define the functions of the components.

```
(%i10)  gsoln:desolve([dex2,dey2,dez2],[x(t),y(t),z(t)])$
Is m q B zero or nonzero?n;
(%i11)  psoln:subst([x(0)=0,y(0)=0,z(0)=0,
             at('diff(x(t),t),t=0)=v0x,
             at('diff(y(t),t),t=0)=v0y,
             at('diff(z(t),t),t=0)=0],gsoln);
```

(psoln) $$[x(t)=t\,v0x,\ y(t)=\frac{m\,v0y\sin\left(\frac{q\,t\,B}{m}\right)}{q\,B},\ z(t)=\frac{m\,v0y\cos\left(\frac{q\,t\,B}{m}\right)}{q\,B}-\frac{m\,v0y}{q\,B}]$$

```
(%i12)  [x(t):=t*v0x,
         y(t):=(m*v0y*sin((q*t*B)/m))/(q*B),
         z(t):=(m*v0y*cos((q*t*B)/m))/(q*B)-(m*v0y)/(q*B)];
```

(%o12) $$[x(t):=t\,v0x,\ y(t):=\frac{m\,v0y\sin\left(\frac{q\,t\,B}{m}\right)}{q\,B},\ z(t):=\frac{m\,v0y\cos\left(\frac{q\,t\,B}{m}\right)}{q\,B}-\frac{m\,v0y}{q\,B}]$$

[$y(t)$, $z(t)$] is the parameter representation of the projection of the path onto the yz-plane. Let's inspect the expressions and try to recognize the curve.

```
(%i13)  y(t)^2+(z(t)+(m*v0y)/(q*B))^2=((m*v0y)/(q*B))^2;
```

(%o13) $$\frac{m^2\,v0y^2\sin\left(\frac{q\,t\,B}{m}\right)^2}{q^2\,B^2}+\frac{m^2\,v0y^2\cos\left(\frac{q\,t\,B}{m}\right)^2}{q^2\,B^2}=\frac{m^2\,v0y^2}{q^2\,B^2}$$

```
(%i14)  trigsimp(%);
```

(%o14) $$\frac{m^2\,v0y^2}{q^2\,B^2}=\frac{m^2\,v0y^2}{q^2\,B^2}$$

Yes, it is a circle of radius $\frac{m\cdot v0y}{q\,B}$ and centre $\left(0,0,-\frac{m\cdot v0y}{q\,B}\right)$. %o14 confirms this.

Below are numerical examples followed by their graphical representation.

```
(%i15)  ex1:subst([m=3,q=-1,v0y=4,B=2],[y(t),z(t)]);
```

(ex1) $$[6\sin\left(\frac{2\,t}{3}\right),\ 6-6\cos\left(\frac{2\,t}{3}\right)]$$

```
(%i18)  ex2:subst([m=3,q=-1,v0y=6,B=2],[y(t),z(t)])$
        ex3:subst([m=3,q=-1,v0y=5,B=-3],[y(t),z(t)])$
        ex4:subst([m=3,q=-1,v0y=6,B=-2],[y(t),z(t)])$
(%i22)  g1:parametric(ex1[1],ex1[2],t,0,10)$
        g2:parametric(ex2[1],ex2[2],t,0,10)$
        g3:parametric(ex3[1],ex3[2],t,0,10)$
        g4:parametric(ex4[1],ex4[2],t,0,10)$
```

```
-->     wxdraw2d(key="v0y=4,B=2",color=red,g1,key="v0y=6,B=2",color=blue,g2,
                 key="v0y=5,B=-3",color=green,g3,
                 key="v0y=6,B=-2",color = black,g4,xlabel="z",ylabel="y")$
```

(%t40)



Now we evaluate and draw the trajectories of an electron.

```
(%i24)  subst([q=-1,m=1,E=0.2,B=-1,v0x=5,v0y=1],[x(t),y(t),z(t)]);
(%o24)  [5 t, sin(t), cos(t)-1]
(%i25)  subst([q=-1,m=1,E=0.2,B=-1,v0x=5,v0y=-1.5],[x(t),y(t),z(t)]);
(%o25)  [5 t, -1.5 sin(t), 1.5-1.5 cos(t)]
(%i26)  subst([q=-1,m=1,E=0.2,B=-1,v0x=5,v0y=1.5],[x(t),y(t),z(t)]);
(%o26)  [5 t, 1.5 sin(t), 1.5 cos(t)-1.5]
(%i29)  spiral1:parametric(5*t,sin(t),cos(t)-1,t,0,20);
        spiral2:parametric(5*t,-1.5*sin(t),1.5-1.5*cos(t),t,0,20);
        spiral3:parametric(5*t,1.5*sin(t),1.5*cos(t)-1.5,t,0,20);
(spiral1) parametric(5 t, sin(t), cos(t)-1, t, 0, 20)
(spiral2) parametric(5 t, -1.5 sin(t), 1.5-1.5 cos(t), t, 0, 20)
(spiral3) parametric(5 t, 1.5 sin(t), 1.5 cos(t)-1.5, t, 0, 20)
(%i30)  draw3d(nticks=200,line_width=2,color=red,view=[80,15],
               spiral1,color=blue,spiral2,color=green,spiral3)$
```



Let us evaluate the time *td* the electron needs to reach the observed plane which has a distance *d* from the initial *yz*-plane

```
(%i31)  solve(x(t)=d,t);
```
(%o31) $\left[t=\dfrac{d}{v0x}\right]$

```
(%i32)  td:d/v0x$
```

The relation between parameters for which $y(td) = 0$ and $z(td) = 0$ can be written in the following compact equation:

```
(%i33)  trigsimp(expand(y(td)^2+z(td)^2))=0;
```
(%o33) $-\dfrac{2\,m^2\,v0y^2\cos\left(\dfrac{d\,q\,B}{m\,v0x}\right)-2\,m^2\,v0y^2}{q^2\,B^2}=0$

There is an equivalent equation in an even simpler form:

$$\cos\left(\frac{d\,q\,B}{m\,v_{0x}}\right)=1\rightarrow\frac{d\,q\,B_k}{m\,v_{0x}}=2k\,\pi,\ \ k \text{ arbitrary integer number.}$$

Comment:

It can easily be seen %o33 that all the electrons are focused at this observed plane (Oscilloscope screen) for an arbitrary velocity value v_{0y}! Consequently the initial velocity in the *yz*-plane has not even to be constant with time.

Now we introduce in this equation the *charge-to-mass* ratio $\left(\dfrac{q}{m}=e\_m\right)$. Then we solve it for *e_m*.

```
(%i34)  eq:subst(q=e_m*m,q*d*B[k]/(m*v0x))=k*2*%pi;
```
(eq) $\dfrac{d\,e\_m\,B_k}{v0x}=2\,\pi\,k$

```
(%i35)  solve(eq,e_m);
```
(%o35) $\left[e\_m=\dfrac{2\,\pi\,k\,v0x}{d\,B_k}\right]$

Electrons are accelerated to the speed v_{0x} by potential U which gives the equation $\dfrac{m\,{v_{0x}}^2}{2}=eU$. The magnetic field B is proportional to the current I: $B_k=\mu_0\dfrac{n}{b}I_k$. We can eliminate v_{0x} and B_k from %o35.

```
(%i36)  subst([v0x=sqrt(2*e_m*U),B[k]=μ[0]*n/b*I[k]],
                e_m=(2*%pi*k*v0x)/(d*B[k]));
```
(%o36) $e\_m=\dfrac{2^{3/2}\,\pi\,b\,k\,\sqrt{e\_m\,U}}{\mu_0\,d\,I_k\,n}$

```
(%i37)  solve(%^2,e_m);
```
b = length of the solenoid and *n* = number of its turns.

(%o37) $\left[e\_m=\dfrac{8\,\pi^2\,b^2\,k^2\,U}{\mu_0^2\,d^2\,I_k^2\,n^2},\ e\_m=0\right]$

V.3 Solve the equations of motion for a particle of mass m and electrical charge q moving in a spatially uniform but time varying electric field of the form $\vec{E} = (E_0 \sin(\omega t), 0, 0)$. Take as initial conditions: $\vec{r}_0 = (x_0, 0, 0)$ and velocity $\vec{v}_0 = (v_0, 0, 0)$.

Solution:

For solving the problem we apply again functions `ode2` and `ic2`. The solution method was presented in the previous problems.

In our case the particle moves along a straight line, therefore *Newton*'s equation reduces to the following form:

```
(%i1)   dex2:m*'diff(x(t),t,2)=q*E[0]*sin(ω*t);
```

(dex2) $m\left(\frac{d^2}{d t^2} x(t)\right) = E_0 \, q \sin(t\,\omega)$

```
(%i2)   ode2(dex2,x(t),t);
```

(%o2) $x(t) = -\frac{E_0 \, q \sin(t\,\omega)}{m\,\omega^2} + \%k2\, t + \%k1$

```
(%i3)   ic2(%,t=0,x(t)=x[0],diff(x(t),t)=v[0]);
```

(%o3) $x(t) = -\frac{E_0 \, q \sin(t\,\omega)}{m\,\omega^2} + \frac{t\,(v_0\, m\,\omega + E_0\, q)}{m\,\omega} + x(0)$

For illustrating the obtained solution we define $x(t)$ as a function and substitute appropriate values for the parameters ($E_0 = 2$, $\omega = 4$, $q = 500$, $m = 3$, $x_0 = 0.5$, $v_0 = 6$).

```
(%i4)   x(t):=-(E[0]*q*sin(t*ω))/
            (m*ω^2)+(t*(v[0]*m*ω+E[0]*q))/(m*ω)+x[0];
```

(%o4) $x(t) := \frac{-E_0 \, q \sin(t\,\omega)}{m\,\omega^2} + \frac{t\,(v_0\, m\,\omega + E_0\, q)}{m\,\omega} + x_0$

```
(%i5)   subst([E[0]=2,q=500,m=3,ω=4,x[0]=0.5,v[0]=6],x(t));
```

(%o5) $-\frac{125 \sin(4\,t)}{6} + \frac{268\,t}{3} + 0.5$

```
(%i6)   plot2d(%,[t,0,5]);
```



V.4 Calculate the trajectory of a particle of mass m and electrical charge q moving in a constant magnetic field $\vec{B}$.
At time $t = 0$ position and velocity of the particle are: $\vec{r}_0 = (0,0,0)$ and $\vec{v}_0 = (v_{0x}, 0, v_{0z})$.

Solution:

Let us assume the co-ordinate system oriented as given in Fig. V.4.



Fig. V.4

Using such a co-ordinate system we can write the field components as follows:

$$B_x = 0,\ B_y = 0,\ B_z = B.$$

For the components of the initial velocity we have:

$$v_{0x} = v_0 \sin(\alpha), v_{0y} = 0,\ v_{0z} = v_0 \cos(\alpha).$$

We enter the definitions of position and velocity vector and of both field vectors.

```
(%i7)   r_(t):=[x(t),y(t),z(t)]$
        vx(t):='diff(x(t),t)$   vy(t):='diff(y(t),t)$
        vz(t):='diff(z(t),z)$
        v_(t):=[vx(t),vy(t),vz(t)]$
        E_:[0,0,0]$   B_:[0,0,B]$
```

We need the library `vector` for applying *Newton*'s equation $\ddot{\vec{r}} = \frac{q}{m}(\vec{v} \times \vec{B})$:

```
(%i7)   r_(t):=[x(t),y(t),z(t)]$
        vx(t):='diff(x(t),t)$   vy(t):='diff(y(t),t)$
        vz(t):='diff(z(t),z)$
        v_(t):=[vx(t),vy(t),vz(t)]$
        E_:[0,0,0]$   B_:[0,0,B]$
(%i8)   load(vect)$
(%i9)   de_:'diff(r_(t),t,2)=q*E_/m+q/m*express(v_(t)~B_);
```

$$\text{(de\_)}\quad \frac{d^2}{d\,t^2}[x(t), y(t), z(t)] = \left[\frac{B\,q\left(\frac{d}{d\,t}\,y(t)\right)}{m}, -\frac{B\,q\left(\frac{d}{d\,t}\,x(t)\right)}{m}, 0\right]$$

```
(%i10)  de_:subst(B=ω*m/q,de_);
```

$$\text{(de\_)}\quad \frac{d^2}{d\,t^2}[x(t), y(t), z(t)] = \left[\left(\frac{d}{d\,t}\,y(t)\right)\omega, -\left(\frac{d}{d\,t}\,x(t)\right)\omega, 0\right]$$

We substitute $\omega = \frac{B \cdot q}{m}$ to get more comfortable expressions and then extract differential Equations for all components.

We extract differential equations for all Cartesian components:

```
(%i11) dex2:'diff(x(t),t,2)=rhs(de_)[1];
```

(dex2) $$\frac{d^2}{d\,t^2}\,x(t)=\left(\frac{d}{d\,t}\,y(t)\right)\,\omega$$

```
(%i12) dey2:'diff(y(t),t,2)=rhs(de_)[2];
```

(dey2) $$\frac{d^2}{d\,t^2}\,y(t)=-\left(\frac{d}{d\,t}\,x(t)\right)\,\omega$$

```
(%i13) dez2:'diff(z(t),t,2)=rhs(de_)[3];
```

(dez2) $$\frac{d^2}{d\,t^2}\,z(t)=0$$

Method 1:

By integration: Solution of DE `dez2` is trivial:

$$z(t=0)=0 \text{ and } \dot{z}(t=0)=v0z \text{ give } \%c2=0, \%c1=v_0\cos(\alpha)\rightarrow z(t)=t\cdot v_0\cos(\alpha).$$

We proceed with calculating $x(t)$ and $y(t)$.

In the first step we integrate one of the two equations `dex2` or `dey2`, let's take the second one. Taking into account the initial condition $\dot{y}(0)=0$, it can easily be seen that constant `%c3` in `%o16` is equal zero. Therefore `%o16` can be written in simpler form %o17.

```
(%i14) integrate(dez2,t);
```

(%o14) $$\frac{d}{d\,t}\,z(t)=\%c1$$

```
(%i15) integrate(%,t);
```

(%o15) $z(t)=\%c1\,t+\%c2$

```
(%i16) integrate(dey2,t);
```

(%o16) $$\frac{d}{d\,t}\,y(t)=\%c3-x(t)\,\omega$$

```
(%i17) 'diff(y(t),t)=-ω*x(t)$
```

In the next step we insert the above derivative into equation `dex2`.

```
(%i18) subst('diff(y(t),t,1)=-x(t)*ω,dex2);
```

(%o18) $$\frac{d^2}{d\,t^2}\,x(t)=-x(t)\,\omega^2$$

It is interesting that `ode2` does not return the correct solution, so I try `desolve` – and it works

```
(%i19) desolve('diff(x(t),t,2)=-omega^2*x(t),x(t));
Is ω zero or nonzero?n;
```

(%o19) $$x(t)=\frac{\sin(\omega\,t)\left(\left.\frac{d}{d\,t}\,x(t)\right|_{t=0}\right)}{\omega}+x(0)\cos(\omega\,t)$$

We can substitute $x'(0)$ and $x(0)$ by copy and past in %o18 or we apply `subst`:

```
(%i20) subst([x(0)=0,at('diff(x(t),t,1),t=0)=v[0]*sin(α)],%);
```

(%o20) $$x(t)=\frac{v_0 \sin(\omega t)\sin(\alpha)}{\omega}$$

Our next step is inserting the obtained function $x(t)$ in expression %i17 followed by integrating to get function $y(t)$:

```
(%i21) subst(x(t)=(v[0]*sin(ω*t)*sin(α))/
              ω,'diff(y(t),t,1)=-ω*x(t));
```

(%o21) $$\frac{d}{d\,t}\,y(t)=-v_0\,\sin(\alpha)\,\sin(t\,\omega)$$

```
(%i22) integrate(%,t);
```

(%o22) $$y(t)=\frac{v_0 \sin(\alpha)\cos(t\,\omega)}{\omega}+\%c4$$

In the final step we apply initial condition $y(0) = 0$ to find constant `%c4`. Then it's easy to get the simplified result for $y(t)$.

```
(%i23) solve(subst(t=0,(v[0]*cos(ω*t)*sin(α)*ω)/ω^2+%c4=0),%c4);
```

(%o23) $$[\,\%c4=-\frac{v_0 \sin(\alpha)}{\omega}\,]$$

```
(%i24) y(t)=(v[0]*cos(ω*t)*sin(α)*ω)/ω^2-(v[0]*sin(α))/ω;
```

(%o24) $$y(t)=\frac{v_0 \sin(\alpha)\cos(t\,\omega)}{\omega}-\frac{v_0 \sin(\alpha)}{\omega}$$

```
(%i25) ratsimp(%);
```

(%o25) $$y(t)=\frac{v_0 \sin(\alpha)\cos(t\,\omega)-v_0 \sin(\alpha)}{\omega}$$

Method 2:

Using complex variables:

The system of equations `[dex2,dey2]` can be solved in an elegant way by introducing a new complex variable

$\eta(t) = x(t) + i \cdot y(t)$ where i denotes the imaginary unit.

We start adding `dex2` and `%i`·`dey2`. Then we rewrite the resulting equation in form of a single equation for the complex function $\eta(t)$.

```
(%i26) B:ω*m/q$
(%i27) dex2+%i*dey2;
```

(%o27) $$\%i\left(\frac{d^2}{d\,t^2}\,y(t)\right)+\frac{d^2}{d\,t^2}\,x(t)=\left(\frac{d}{d\,t}\,y(t)\right)\omega-\%i\left(\frac{d}{d\,t}\,x(t)\right)\omega$$

```
(%i28) diff(η(t),t,2)=-%i*ω*diff(eta(t),t);
```

(%o28) $$\frac{d^2}{d\,t^2}\,\eta(t)=-\%i\left(\frac{d}{d\,t}\,\eta(t)\right)\omega$$

At first we perform the integration:

```
(%i29)  integrate(%,t);
```

(%o29) $\frac{d}{d\,t}\,\eta(t) = \%c5 - \%i\,\eta(t)\,\omega$

Taking into account the initial conditions

$$x(0)=0,\ \ y(0)=0,\ \ \dot{x}(0)=v_0\sin(\alpha),\ \ \dot{y}(0)=0$$

we get

$$\dot{\eta}(0)=\dot{x}(0)+i\,\dot{y}(0)=v_0\sin(\alpha)$$

and equation %o29 can be written as

```
(%i30)  'diff(η(t),t)=v[0]*sin(α)-%i*(q*η(t)*B)/m;
```

(%o30) $\frac{d}{d\,t}\,\eta(t) = v_0\,\sin(\alpha) - \%i\,\eta(t)\,\omega$

We apply ode2 - which can be applied for 1st order DEs, too – and define $\eta(t)$.

```
(%i31)  ode2(%,η(t),t);
```

(%o31) $\eta(t) = \%e^{-\%i\,t\,\omega}\left(\%c - \frac{\%i\,v_0\,\sin(\alpha)\,\%e^{\%i\,t\,\omega}}{\omega}\right)$

```
(%i32)  η(t):=%e^(-%i*t*ω)*(%c-(%i*v[0]*sin(α)*%e^(%i*t*ω))/ω);
```

(%o32) $\eta(t) := \%e^{(-\%i)\,t\,\omega}\left(\%c - \frac{\%i\,v_0\,\sin(\alpha)\,\%e^{\%i\,t\,\omega}}{\omega}\right)$

Considering the initial condition delivers constant %c.

```
(%i33)  solve(η(0)=0,%c);
```

(%o33) $[\ \%c = \frac{\%i\,v_0\,\sin(\alpha)}{\omega}\ ]$

```
(%i34)  η(t):=%e^((-%i)*t*ω)*
           ((%i*v[0]*sin(α))/ω-(%i*v[0]*sin(α)*%e^(%i*t*ω))/ω);
```

(%o34) $\eta(t) := \%e^{(-\%i)\,t\,\omega}\left(\frac{\%i\,v_0\,\sin(\alpha)}{\omega} - \frac{\%i\,v_0\,\sin(\alpha)\,\%e^{\%i\,t\,\omega}}{\omega}\right)$

```
(%i35)  ratsimp(%);
```

(%o35) $\eta(t) := -\frac{\%i\,v_0\,\sin(\alpha)\,\%e^{\%i\,t\,\omega} - \%i\,v_0\,\sin(\alpha)}{\omega\,\%e^{\%i\,t\,\omega}}$

```
(%i36)  η_(t):=expand(η(t));
```

(%o36) $\eta(t) := \text{expand}(\eta(t))$

```
(%i37)  x(t)=realpart(η_(t));
```

(%o37) $x(t) = \frac{v_0\,\sin(\alpha)\,\sin(t\,\omega)}{\omega}$

```
(%i38)  y(t)=imagpart(η_(t));
```

(%o38) $y(t) = \frac{v_0\,\sin(\alpha)\,\cos(t\,\omega)}{\omega} - \frac{v_0\,\sin(\alpha)}{\omega}$

Some manipulations give the final form of $\eta(t)$.

It remains to extract real and imaginary part in order to obtain requested functions $x(t)$ and $y(t)$.

Method 3: Using `desolve`.

```
(%i39)  gsoln:desolve([dex2,dey2,dez2],[x(t),y(t),z(t)])$
Is ω zero or nonzero?n;
(%i40)  psoln:subst([x(0)=0,y(0)=0,z(0)=0,
                  at('diff(x(t),t),t=0)=v[0]*sin(α),
                  at('diff(y(t),t),t=0)=0,
                  at('diff(z(t),t),t=0)=v[0]*cos(α)],gsoln);
```

(psoln) $\left[x(t)=\frac{v_0\sin(\alpha)\sin(t\,\omega)}{\omega},\ y(t)=\frac{v_0\sin(\alpha)\cos(t\,\omega)}{\omega}-\frac{v_0\sin(\alpha)}{\omega},\ z(t)=v_0\,t\cos(\alpha)\right]$

```
(%i42)  x(t):=rhs(psoln[1])$  x(t);
```

(%o42) $\frac{v_0\sin(\alpha)\sin(t\,\omega)}{\omega}$

```
(%i44)  y(t):=expand(rhs(psoln[2]))$  y(t);
```

(%o44) $\frac{v_0\sin(\alpha)\cos(t\,\omega)}{\omega}-\frac{v_0\sin(\alpha)}{\omega}$

```
(%i46)  z(t):=expand(rhs(psoln[3]))$  z(t);
```

(%o46) $v_0\,t\cos(\alpha)$

Comparing the results we fortunately can observe that they are the same.

Finally we will resubstitute for $\omega=\frac{B\cdot q}{m}$.

```
(%i47)  kill(B)$
(%i49)  x_(t):=subst(ω=B*q/m,x(t))$  x_(t);
```

(%o49) $\frac{v_0\,m\sin\left(\frac{B\,q\,t}{m}\right)\sin(\alpha)}{B\,q}$

```
(%i51)  y_(t):=subst(ω=B*q/m,y(t))$  y_(t);
```

(%o51) $\frac{v_0\,m\cos\left(\frac{B\,q\,t}{m}\right)\sin(\alpha)}{B\,q}-\frac{v_0\,m\sin(\alpha)}{B\,q}$

Now having done all the work we would like to see the trajectory of the particle for one data set:

```
(%i52)  subst([v[0]=5,α=%pi/3,B=3,m=2,q=1],[x_(t),y_(t),z(t)]);
```

$$(\%o52)\quad \left[\frac{5\sin\left(\frac{3t}{2}\right)}{\sqrt{3}},\frac{5\cos\left(\frac{3t}{2}\right)}{\sqrt{3}}-\frac{5}{\sqrt{3}},\frac{5t}{2}\right]$$

```
(%i53)  spiral:parametric((5*sin((3*t)/2))/sqrt(3),
                          (5*cos((3*t)/2))/sqrt(3)-5/sqrt(3),
                          (5*t)/2,t,0,3*%pi);
```

$$(\text{spiral})\quad \text{parametric}\left(\frac{5\sin\left(\frac{3t}{2}\right)}{\sqrt{3}},\frac{5\cos\left(\frac{3t}{2}\right)}{\sqrt{3}}-\frac{5}{\sqrt{3}},\frac{5t}{2},t,0,3\pi\right)$$

```
-->     wxdraw3d(key="Particle Motion",
              nticks=200,line_width=2,color=blue,spiral)$
```

(%t55)



Some years ago we solved all problems using DERIVE. It might be interesting for the reader to compare the solving procedures from now and then.

We think that the code doesn't need too many explanations.

$$\#1: \quad [\text{InputMode} := \text{Word},\ \text{CaseMode} := \text{Sensitive}]$$

$$\#2: \quad [x(t) :=,\ y(t) :=,\ z(t) :=]$$

$$\#3: \quad r\_(t) := [x(t),\ y(t),\ z(t)]$$

$$\#4: \quad v\_(t) := \frac{d}{dt}\ r\_(t)$$

$$\#5: \quad [B\_ := [0,\ 0,\ B],\ r0\_ := [0,\ 0,\ 0]]$$

$$\#6: \quad v0\_ := [v0\cdot \text{SIN}(\alpha),\ 0,\ v0\cdot \text{COS}(\alpha)]$$

$$\#7: \quad eq := \frac{d}{dt}\ \frac{d}{dt}\ r\_(t) = \frac{q}{m}\cdot \text{CROSS}(v\_(t),\ B\_)$$

$$\#8: \quad eq := [x''(t),\ y''(t),\ z''(t)] = \left[\frac{B\cdot q\cdot y'(t)}{m},\ -\frac{B\cdot q\cdot x'(t)}{m},\ 0\right]$$

$$\#9: \quad eq := x''(t) = \frac{B\cdot q\cdot y'(t)}{m} \wedge y''(t) = -\frac{B\cdot q\cdot x'(t)}{m} \wedge z''(t) = 0$$

$$\#10: \quad eq\_ := \left[x''(t) = \frac{B\cdot q\cdot y'(t)}{m},\ y''(t) = -\frac{B\cdot q\cdot x'(t)}{m},\ z''(t) = 0\right]$$

Method 1:

$$\#11: \quad \int eq\_\ dt = \left[x'(t) = \frac{B\cdot q\cdot y(t)}{m},\ y'(t) = -\frac{B\cdot q\cdot x(t)}{m},\ z'(t) = 0\right]$$

$$\#12: \quad eq1 := \left[x'(t) = v0\_{}_1 + \frac{B\cdot q\cdot y(t)}{m},\ y'(t) = v0\_{}_2 - \frac{B\cdot q\cdot x(t)}{m},\ z'(t) = v0\_{}_3\right]$$

$$\#13: \quad eq1 := \left[x'(t) = v0\cdot \text{SIN}(\alpha) + \frac{B\cdot q\cdot y(t)}{m},\ y'(t) = -\frac{B\cdot q\cdot x(t)}{m},\ z'(t) = v0\cdot \text{COS}(\alpha)\right]$$

$$\#14: \quad x''(t) = \frac{B\cdot q\cdot \text{RHS}(eq1_2)}{m}$$

$$\#15: \quad x''(t) = -\frac{B^2\cdot q^2\cdot x(t)}{m^2}$$

$$\#16: \quad y''(t) = B\cdot q\cdot \text{RHS}(eq1_1)$$

$$\#17: \quad y''(t) = B\cdot q\cdot v0\cdot \text{SIN}(\alpha) + \frac{B^2\cdot q^2\cdot y(t)}{m}$$

For solving the DEs #15 and #17 we will use the DERIVE tool DSOLVE2_IV, which needs a special syntax:

We must bring the equations in the form: $\ddot{x}(t) + p\cdot\dot{x}(t) + q\cdot x(t) = r(x)$.

Then we can apply $\text{DSOLVE\_IV}(p,q,r,t,t_0,x(t_0),\dot{x}(t_0))$. We do this for both equations:

$$\#18:\quad x(t) := \text{DSOLVE2\_IV}\left(0,\ \frac{B^2\cdot q^2}{m^2},\ 0,\ t,\ 0,\ 0,\ v0\_{}_1\right)$$

$$\#19:\quad x(t) := \frac{m\cdot v0\cdot \text{SIN}(\alpha)\cdot \text{SIN}\left(\frac{B\cdot q\cdot t}{m}\right)}{B\cdot q}$$

$$\#20:\quad y(t) := \text{DSOLVE2\_IV}\left(0,\ \frac{B^2\cdot q^2}{m^2},\ -\frac{B\cdot q\cdot v0\_{}_1}{m},\ t,\ 0,\ 0,\ 0\right)$$

$$\#21:\quad y(t) := \frac{m\cdot v0\cdot \text{SIN}(\alpha)\cdot \text{COS}\left(\frac{B\cdot q\cdot t}{m}\right)}{B\cdot q} - \frac{m\cdot v0\cdot \text{SIN}(\alpha)}{B\cdot q}$$

$$\#22:\quad y(t) := \frac{m\cdot v0\cdot \text{SIN}(\alpha)\cdot\left(\text{COS}\left(\frac{B\cdot q\cdot t}{m}\right) - 1\right)}{B\cdot q}$$

We can define the trajectory putting all components together, substitute the provided data and plot the curve.

$$\#23:\quad \left[x(t),\ y(t),\ t\cdot v0\_{}_3\right]$$

$$\#24:\quad \left[\frac{m\cdot v0\cdot \text{SIN}(\alpha)\cdot \text{SIN}\left(\frac{B\cdot q\cdot t}{m}\right)}{B\cdot q},\ \frac{m\cdot v0\cdot \text{SIN}(\alpha)\cdot \text{COS}\left(\frac{B\cdot q\cdot t}{m}\right)}{B\cdot q} - \frac{m\cdot v0\cdot \text{SIN}(\alpha)}{B\cdot q},\ v0\cdot t\cdot \text{COS}(\alpha)\right]$$

$$\#25:\quad \left[\frac{5\cdot\sqrt{3}\cdot \text{SIN}\left(\frac{3\cdot t}{2}\right)}{3},\ \frac{5\cdot\sqrt{3}\cdot \text{COS}\left(\frac{3\cdot t}{2}\right)}{3} - \frac{5\cdot\sqrt{3}}{3},\ \frac{5\cdot t}{2}\right]$$



Method 2

#26: [x(t) :=, y(t) :=, z(t) :=]

#27: $\eta(t) := x(t) + \hat{\imath}\cdot y(t)$

#28: $x''(t) = x''(t)$

#29: $$x''(t) + y''(t) = \frac{B\cdot q\cdot y'(t)}{m} + \hat{\imath}\cdot\left(-\frac{B\cdot q\cdot x'(t)}{m}\right)$$

#30: $$x''(t) + y''(t) = -\frac{B\cdot q\cdot(-y'(t) + \hat{\imath}\cdot x'(t))}{m}$$

#31: $$\left(\frac{d}{dt}\right)^2 \eta(t) = -\frac{B\cdot q}{m}\cdot\eta(t)$$

We can define the trajectory putting all components together, substitute the provided data and plot the curve.

Then there was given the hint to proceed applying DSOLVE and the remainder of the exercise was left to the reader to complete.

Method 3 could not be applied using DERIVE but there was a screen shot of a *Mathematica*-procedure applying DSolve:

```
{Bz = B, By = 0, Bx = 0, xo = 0, yo = 0, zo = 0, vox = vo Cos[α],
  voy = 0, voz = vo Sin[α]};
DSolve[{m x''[t] == q (y'[t] Bz - By z'[t]),
        m y''[t] == q (z'[t] Bx - Bz x'[t]),
        m z''[t] == q (x'[t] By - Bx y'[t]),
        x[0] == xo, y[0] == yo, z[0] == zo,
        x'[0] == vox, y'[0] == voy, z'[0] == voz},
        {x[t], y[t], z[t]}, t]
```

You are invited to compare with Method 3 performed with Maxima.

But there was also a small add-on:

It was shown that the kinetic energy of the particle is constant.

(The components of *r_*(*t*) are the final results of the calculation above.

#33: $$Ek(t) := \frac{m}{2}\cdot\left(\frac{d}{dt}\ r\_(t)\right)^2$$

#34: $$Ek(t) := \frac{m\cdot v0^2}{2}$$

It is clear that the kinetic energy depends only on the initial velocity v_0, i.e. it remains constant.

Note: The result obtained is the result we could have expected because the magnetic force is always perpendicular to the displacement, which means that it does not do any work.

V.5 Calculate the trajectory of a particle of mass m and electrical charge q moving in a constant electric and magnetic field which are mutually perpendicular.

Solution:

Let us assume the co-ordinate system oriented as given in Fig. V.5.



Fig. V.5

Using such a coordinate system we have

$$\vec{E} = (E,0,0) \text{ and } \vec{B} = (0,0,B).$$

Let's assume the following initial conditions:

$$x_0 = 0, y_0 = 0, z_0 = 0 \text{ and } \vec{v}_0 = (v_{0x}, v_{0y}, v_{0z}).$$

We enter position vector and velocity vector followed by the field vectors and proceed like in the problems above.

```
(%i7)   r_(t):=[x(t),y(t),z(t)]$
        vx(t):='diff(x(t),t)$  vy(t):='diff(y(t),t)$
        vz(t):='diff(z(t),z)$
        v_(t):=[vx(t),vy(t),vz(t)]$
        E_:[E,0,0]$  B_:[0,0,B]$
(%i8)   load(vect)$
(%i9)   de_:'diff(r_(t),t,2)=q*E_/m+q/m*express(v_(t)~B_);
```

(de_) $$\frac{d^2}{d\,t^2}\,[x(t), y(t), z(t)] = \left[\frac{B\,q\left(\frac{d}{d\,t}\,y(t)\right)}{m} + \frac{E\,q}{m}, -\frac{B\,q\left(\frac{d}{d\,t}\,x(t)\right)}{m}, 0\right]$$

```
(%i10)  dex2:'diff(x(t),t,2)=rhs(de_)[1];
```

(dex2) $$\frac{d^2}{d\,t^2}\,x(t) = \frac{B\,q\left(\frac{d}{d\,t}\,y(t)\right)}{m} + \frac{E\,q}{m}$$

```
(%i11)  dey2:'diff(y(t),t,2)=rhs(de_)[2];
```

(dey2) $$\frac{d^2}{d\,t^2}\,y(t) = -\frac{B\,q\left(\frac{d}{d\,t}\,x(t)\right)}{m}$$

```
(%i12)  dez2:'diff(z(t),t,2)=rhs(de_)[3];
```

(dez2) $$\frac{d^2}{d\,t^2}\,z(t) = 0$$

Below are all components of the particular solution describing the trajectory.

```
(%i14) psoln:subst([x(0)=0,y(0)=0,z(0)=0,
              at('diff(x(t),t),t=0)=v0x,
              at('diff(y(t),t),t=0)=v0y,
              at('diff(z(t),t),t=0)=v0z],gsoln)$
(%i15) psoln[1];
```

(%o15) $$x(t)=\frac{\frac{m^3\sin\left(\frac{B\,q\,t}{m}\right)v0x}{B\,q}-\frac{\cos\left(\frac{B\,q\,t}{m}\right)\left(B\,m^3\,v0y+E\,m^3\right)}{B^2\,q}}{m^2}+\frac{B\,m\,v0y+E\,m}{B^2\,q}$$

```
(%i16) psoln[2];
```

(%o16) $$y(t)=\frac{\frac{m\sin\left(\frac{B\,q\,t}{m}\right)\left(B\,m^2\,v0y+E\,m^2\right)}{B^2\,q}+\frac{m^3\cos\left(\frac{B\,q\,t}{m}\right)v0x}{B\,q}}{m^2}-\frac{m\,v0x}{B\,q}-\frac{E\,t}{B}$$

```
(%i17) psoln[3];
```

(%o17) $z(t)=t\,v0z$

We plot the space curve for a chosen data set:

```
(%i18) x(t):=rhs(psoln[1])$
(%i19) y(t):=rhs(psoln[2])$
(%i20) z(t):=rhs(psoln[3])$
(%i21) subs:[E=3,B=2,m=1,q=-1,
         v0x=v0*sin(α),v0y=v0*cos(α),v0z=0.5,v0=2,α=%pi/4]$
(%i24) xspec(t):=subst(subs,x(t))$  yspec(t):=subst(subs,y(t))$
       zspec(t):=subst(subs,z(t))$
(%i25) tra:parametric(xspec(t),yspec(t),zspec(t),t,0,6*%pi)$
(%i28) load(draw)$
0 errors, 0 warnings
(%i27) draw3d(key="Trajectory",nticks=200,line_width=2,
          color = blue,tra);
(%o27) [gr3d(parametric)]
```



V.6 Calculate the trajectory of a particle of mass m and electrical charge q moving in a constant electric and magnetic field which are oriented in opposite direction to each other.

Take this problem as an exercise.

You may assume that the co-ordinate system is oriented as given in Fig. V.6.



Fig. V.6

Electric and magnetic field are given as $\vec{E} = (-E, 0, 0)$ and $\vec{B} = (B, 0, 0)$.

Assume the following initial conditions: $x_0 = 0,\ y_0 = 0,\ z_0 = 0;\ \vec{v}(v_{0x}, v_{0y}, 0)$

Solution:

The equation of the trajectory is

$$tr(t) = \left(t\, v_{0x} - \frac{q t^2 E}{2m}, \frac{m v_{0y}}{q B} \sin \frac{q t B}{m}, \frac{m v_{0y}}{q B} \left(\cos \frac{q t B}{m} - 1 \right) \right).$$

See a possible trajectory:



V.7 Solve the equations of motion for a charged particle of mass m and charge q moving in an oscillating electric field $\vec{E}=(E\cos(\omega t),-E\sin(\omega t),0)$, where $\omega=\frac{qB}{m}$, and a uniform and constant magnetic field oriented perpendicular to this electric field $\vec{B}=(0,0,B)$. Take as initial conditions $\vec{r}_0=(0,0,0), \vec{v}_0=(0,0,0)$.

Solution:

We can work as before and apply the third method.

```
(%i1)   kill(all);
(%o0)   done
(%i7)   r_(t):=[x(t),y(t),z(t)]$
        vx(t):='diff(x(t),t)$   vy(t):='diff(y(t),t)$
        vz(t):='diff(z(t),z)$
        v_(t):=[vx(t),vy(t),vz(t)]$
        E_:[E*cos(ω*t),-E*sin(ω*t),0]$   B_:[0,0,B]$
(%i8)   load(vect)$
(%i9)   de_:diff(r_(t),t,2)=q*E_/m+q/m*express(v_(t)~B_)$
(%i10)  dex2:lhs(de_)[1]=rhs(de_)[1];
```

(dex2) $$\frac{d^2}{d t^2}\,x(t)=\frac{E\,q\cos(t\,\omega)}{m}+\frac{B\,q\left(\frac{d}{d\,t}\,y(t)\right)}{m}$$

```
(%i11)  dey2:lhs(de_)[2]=rhs(de_)[2];
```

(dey2) $$\frac{d^2}{d t^2}\,y(t)=-\frac{E\,q\sin(t\,\omega)}{m}-\frac{B\,q\left(\frac{d}{d\,t}\,x(t)\right)}{m}$$

```
(%i12)  dez2:lhs(de_)[3]=rhs(de_)[3];
```

(dez2) $$\frac{d^2}{d t^2}\,z(t)=0$$

```
(%i13)  gsoln:desolve([dex2,dey2,dez2],[x(t),y(t),z(t)])$
Is ω zero or nonzero?n;
  Is B m q zero or nonzero?n;
(%i14)  psoln:subst([x(0)=0,y(0)=0,z(0)=0,
                     at('diff(x(t),t),t=0)=0,
                     at('diff(y(t),t),t=0)=0,
                     at('diff(z(t),t),t=0)=0],gsoln)$
```

We receive the components of the trajectory in explicit form and define them as functions:

```
(%i15)  psoln[1];
```

(%o15) $$x(t)=-\frac{E\,q\cos(t\,\omega)}{m\,\omega^2-B\,q\,\omega}+\frac{E\,m\,q\cos\left(\frac{B\,q\,t}{m}\right)}{B\,m\,q\,\omega-B^2\,q^2}-\frac{E}{B\,\omega}$$

```
(%i16)  x(t):=-(E*q*cos(t*ω))/(m*ω^2-B*q*ω)+
            (E*m*q*cos((B*q*t)/m))/(B*m*q*ω-B^2*q^2)-E/(B*ω)$
```

```
(%i17)  psoln[2];
```

$$\text{(\%o17)}\quad y(t)=\frac{E\,q\sin(t\,\omega)}{\omega\,(m\,\omega-B\,q)}-\frac{E\,m\sin\left(\frac{B\,q\,t}{m}\right)}{B\,(m\,\omega-B\,q)}$$

```
(%i18)  y(t):=(E*q*sin(t*ω))/(ω*(m*ω-B*q))-
             (E*m*sin((B*q*t)/m))/(B*(m*ω-B*q))$
(%i19)  psoln[3];
```

$$\text{(\%o19)}\quad z(t)=0$$

```
(%i20)  z(t):=0$
(%i21)  limit([x(t),y(t)],B,(m*ω)/q);
```

$$\text{(\%o21)}\quad [-\frac{E\left(-q\,t\,\omega\sin(t\,\omega)-q\cos(t\,\omega)+q\right)}{m\,\omega^2},-\frac{E\,q\sin(t\,\omega)-E\,q\,t\,\omega\cos(t\,\omega)}{m\,\omega^2}]$$

```
(%i22)  subst([B=(m*ω)/q],[x(t),y(t)]);
expt: undefined: 0 to a negative exponent.
  -- an error. To debug this try: debugmode(true);
```

As you can see it is necessary to calculate the limit, substitution does not work because the denominator of $x(t$ and $y(t)$ equals zero.

Finally we will plot the trajectory based on a set of parameters.

```
(%i23)  xx(t):=subst([E=2,q=-1,m=3,ω=10],
             -(E*(-q*t*ω*sin(t*ω)-q*cos(t*ω)+q))/(m*ω^2))$
        xx(t);
```

$$\text{(\%o23)}\quad -\frac{10\,t\sin(10\,t)+\cos(10\,t)-1}{150}$$

```
(%i25)  yy(t):=subst([E=2,q=-1,m=3,ω=10],
             -(E*q*sin(t*ω)-E*q*t*ω*cos(t*ω))/(m*ω^2))$
        yy(t);
```

$$\text{(\%o25)}\quad \frac{2\sin(10\,t)-20\,t\cos(10\,t)}{300}$$

```
(%i26)  traj:parametric(xx(t),yy(t),z(t),t,0,2*%pi)$
-->     wxdraw3d(key="Trajectory",nticks=400,line_width=2,
             color=red,traj)$
```



We will demonstrate how to perform "Method 1" with TI-NspireCAS:

$$r\_(t):=[x(t)\ \ y(t)\ \ z(t)]:v\_(t):=\frac{d}{dt}(r\_(t)) \qquad \textit{Done}$$

$$bb\_:=\begin{bmatrix}0 & 0 & \frac{m\cdot\omega}{q}\end{bmatrix}:ee\_:=[ee\cdot\cos(\omega\cdot t)\ \ {-ee}\cdot\sin(\omega\cdot t)\ \ 0] \qquad [ee\cdot\cos(\omega\cdot t)\ \ {-ee}\cdot\sin(\omega\cdot t)\ \ 0]$$

$$eq:=m\cdot\frac{d^2}{dt^2}(r\_(t))=q\cdot ee\_+q\cdot\text{crossP}(v\_(t),bb\_)$$

$$\left[m\cdot\frac{d^2}{dt^2}(x(t))=m\cdot\omega\cdot\frac{d}{dt}(y(t))+q\cdot ee\cdot\cos(\omega\cdot t)\ \ \ m\cdot\frac{d^2}{dt^2}(y(t))=-m\cdot\omega\cdot\frac{d}{dt}(x(t))-q\cdot ee\cdot\sin(\omega\cdot t)\ \ \ m\cdot\frac{d^2}{dt^2}(z(t))=0\right]$$

$$de1:=\frac{\int\left(m\cdot\frac{d^2}{dt^2}(x(t))\right)dt=\int\left(m\cdot\omega\cdot\frac{d}{dt}(y(t))+q\cdot ee\cdot\cos(\omega\cdot t)\right)dt}{m} \qquad \frac{d}{dt}(x(t))=\frac{q\cdot ee\cdot\sin(\omega\cdot t)+m\cdot\omega^2\cdot y(t)}{m\cdot\omega}$$

$$de2:=\frac{\int\left(m\cdot\frac{d^2}{dt^2}(y(t))\right)dt=\int\left(-m\cdot\omega\cdot\frac{d}{dt}(x(t))-q\cdot ee\cdot\sin(\omega\cdot t)\right)dt}{m} \qquad \frac{d}{dt}(y(t))=\frac{q\cdot ee\cdot\cos(\omega\cdot t)-m\cdot\omega^2\cdot x(t)}{m\cdot\omega}$$

$$de3:=\frac{\int\left(m\cdot\frac{d^2}{dt^2}(z(t))\right)dt=\int 0\,dt}{m} \qquad \frac{d}{dt}(z(t))=0$$

Short comments are provided.

$$de3:=\frac{\int\left(m\cdot\frac{d^2}{dt^2}(z(t))\right)dt=\int 0\,dt}{m} \qquad \frac{d}{dt}(z(t))=0$$

© Integration constant for de2:

$$\text{solve}\left(0=\frac{q\cdot ee\cdot\cos(\omega\cdot 0)-m\cdot\omega^2\cdot 0}{m\cdot\omega}+c,c\right) \qquad c=\frac{-q\cdot ee}{m\cdot\omega}$$

$$\frac{d}{dt}(y(t))=\frac{q\cdot ee\cdot\cos(\omega\cdot t)-m\cdot\omega^2\cdot x(t)}{m\cdot\omega}+\frac{-q\cdot ee}{m\cdot\omega} \qquad \frac{d}{dt}(y(t))=\frac{q\cdot ee\cdot\cos(\omega\cdot t)-m\cdot\omega^2\cdot x(t)-q\cdot ee}{m\cdot\omega}$$

© Substitute in 1st de for y':

$$m\cdot\frac{d^2}{dt^2}(x(t))=m\cdot\omega\cdot\frac{q\cdot ee\cdot\cos(\omega\cdot t)-m\cdot\omega^2\cdot x(t)-q\cdot ee}{m\cdot\omega}+q\cdot ee\cdot\cos(\omega\cdot t) \qquad m\cdot\frac{d^2}{dt^2}(x(t))=2\cdot q\cdot ee\cdot\cos(\omega\cdot t)-m\cdot\omega^2\cdot x(t)-q\cdot ee$$

$$\text{deSolve}\left(m\cdot x''=2\cdot q\cdot ee\cdot\cos(\omega\cdot t)-m\cdot\omega^2\cdot x-q\cdot ee \text{ and } x(0)=0 \text{ and } x'(0)=0,t,x\right)|\omega>0$$

$$x=\frac{-q\cdot ee\cdot\left((\cos(\omega\cdot t))^2-\cos(\omega\cdot t)+\sin(\omega\cdot t)\cdot(\sin(\omega\cdot t)-\omega\cdot t)\right)}{m\cdot\omega^2} \text{ and } \omega>0$$

$$x(t):=\text{tCollect}\left(\frac{-q\cdot ee\cdot\left((\cos(\omega\cdot t))^2-\cos(\omega\cdot t)+\sin(\omega\cdot t)\cdot(\sin(\omega\cdot t)-\omega\cdot t)\right)}{m\cdot\omega^2}\right) \qquad \textit{Done}$$

$m\cdot \frac{d^2}{dt^2}(x(t))=m\cdot \omega\cdot \frac{q\cdot ee\cdot \cos(\omega\cdot t)-m\cdot \omega^2\cdot x(t)-q\cdot ee}{m\cdot \omega}+q\cdot ee\cdot \cos(\omega\cdot t)$ $\quad m\cdot \frac{d^2}{dt^2}(x(t))=2\cdot q\cdot ee\cdot \cos(\omega\cdot t)-m\cdot \omega^2\cdot x(t)-q\cdot ee$

deSolve$\left(m\cdot x''=2\cdot q\cdot ee\cdot \cos(\omega\cdot t)-m\cdot \omega^2\cdot x-q\cdot ee \text{ and } x(0)=0 \text{ and } x'(0)=0,t,x\right)|\omega>0$

$$x=\frac{-q\cdot ee\cdot \left((\cos(\omega\cdot t))^2-\cos(\omega\cdot t)+\sin(\omega\cdot t)\cdot (\sin(\omega\cdot t)-\omega\cdot t)\right)}{m\cdot \omega^2} \text{ and } \omega>0$$

$x(t):=\text{tCollect}\left(\frac{-q\cdot ee\cdot \left((\cos(\omega\cdot t))^2-\cos(\omega\cdot t)+\sin(\omega\cdot t)\cdot (\sin(\omega\cdot t)-\omega\cdot t)\right)}{m\cdot \omega^2}\right)$ *Done*

$x(t)$ $\quad \frac{q\cdot ee\cdot (\cos(\omega\cdot t)+\omega\cdot t\cdot \sin(\omega\cdot t)-1)}{m\cdot \omega^2}$

$y(t):=\int \frac{q\cdot ee\cdot \cos(\omega\cdot t)-m\cdot \omega^2\cdot x(t)-q\cdot ee}{m\cdot \omega}\,dt$ *Done*

$x1(t):=x(t):y1(t):y(t)$ $\quad \frac{q\cdot ee\cdot (\omega\cdot t\cdot \cos(\omega\cdot t)-\sin(\omega\cdot t))}{m\cdot \omega^2}$

$x1(t):=x(t):y1(t):=y(t)$ *Done*

We plot the projection of the space curve onto the *xy*-plane.





V.8 Investigate the motion of a particle α moving from infinity towards the nucleus of an atom with Z protons. Assume that the mass of particle m is much smaller than the mass of the nucleus M.

Solution:

Total energy of the particle E_n, which is the sum of the kinetic energy and the potential energy is constant. Thus

$$E_k + E_p = E_n = const \qquad (*)$$

In polar coordinates the energies are given by:

$$E_k = \frac{1}{2}m(\dot{r}^2 + r^2\dot{\varphi}^2), \qquad E_p = \frac{2Ze^2}{r} \quad (e = \text{elementary charge})$$

Assuming that the mass of the particle is much smaller than the mass of the nucleus we can consider the position of the nucleus as a fixed one. Thus, the area mapped out per second by the position vector is constant:

$$r^2\dot{\varphi} = C.$$

On the other hand:

$$\frac{dr}{dt} = \frac{dr}{d\varphi}\frac{d\varphi}{dt} = \frac{dr}{d\varphi}\frac{C}{r^2}.$$

Hence, substituting the above relations into the equation for the conservation of energy (*) s to the equation

$$\frac{m}{2}\left(\frac{C^2}{r^4}\left(\frac{dr}{d\varphi}\right)^2 + \frac{C^2}{r^2}\right) + \frac{2Ze^2}{r} = E_n \qquad (**)$$

We introduce a new variable: $\rho(\varphi) = \dfrac{1}{r(\varphi)}$,

calculate the derivative

$$\frac{dr(\varphi)}{d\varphi} = -r^2(\varphi)\frac{d\rho(\varphi)}{d\varphi} = -\frac{1}{\rho^2(\varphi)}\frac{d\rho(\varphi)}{d\varphi}$$

and substitute both into the equation (**):

$$\left(\frac{d\rho(\varphi)}{d\varphi}\right)^2 + \rho^2(\varphi) + A\rho(\varphi) = B \quad \text{where } A = \frac{4Ze^2}{mC^2},\ B = \frac{2E_n}{mC^2}.$$

Hence, the equation for the conversion of the energy is now in a simpler form.

Differentiating this equation and cancelling the factor $2 \cdot \dfrac{d\rho(\varphi)}{d\varphi}$, we obtain the following equation:

$$\frac{d^2\rho(\varphi)}{d\varphi^2} + \rho(\varphi) + \frac{A}{2} = 0.$$

We will solve this equation supported by the `ode2` function:

```
(%i1)   eq:'diff(ρ,φ,2)+ρ+A/2=0$
(%i2)   ode2(eq,ρ,φ);
(%o2)   ρ= %k1 sin(φ)+ %k2 cos(φ) - A/2
(%i3)   ρ(φ):=%k1*sin(φ)+%k2*cos(φ)-A/2$
```

The constants `%k1`, `%k2` can be calculated from the initial conditions as follows. The trajectory of the particle at infinity is of course just a straight line. Suppose that this straight line, when extrapolated, passes the nucleus at a distance d. This means that at time $t = 0$, we can write the following two initial conditions (see Fig. V.8):

$$r(\varphi)\big|_{\varphi=0} = \infty \rightarrow \rho(\varphi)\big|_{\varphi=\pi} = 0$$

and

$$y(\varphi) = r(\varphi)\sin(\varphi) \rightarrow y(\pi) = \lim_{\varphi\to\pi}\frac{\sin(\varphi)}{\rho(\varphi)} = d.$$



Fig. V.8

```
(%i4)   solve(ρ(%pi)=0,%k2);
(%o4)   [ %k2=- A/2 ]
(%i5)   %k2:-A/2$
```

%k2 was easy to find. %k1 can be found from the second initial condition. Then we can plot the trajectory:

```
(%i8)   r(φ):=1/ρ(φ)$ x(φ):=r(φ)*cos(φ)$ y(φ):=r(φ)*sin(φ)$
(%i9)   limit(y(φ),φ,%pi)=d;
```

(%o9) $\frac{1}{\%k1}=d$

```
(%i10)  %k1:1/d$
(%i11)  g1: subst([d=1/20,A=20],parametric(x(φ),y(φ), φ,0,%pi));
```

(g1) $\mathrm{parametric}\left(\frac{\cos(\varphi)}{20\sin(\varphi)-10\cos(\varphi)-10},\frac{\sin(\varphi)}{20\sin(\varphi)-10\cos(\varphi)-10},\varphi,0,\pi\right)$

```
(%i12)  text:label(["trajectory for A=20,d=0.05",-0.4,0.15])$
-->     wxdraw2d(nticks=200,color=red,
            xaxis=true,yaxis=true,
            xrange=[-1,0.5],yrange=[0,0.2],
            title="y(x)",g1,text)$
```

(%t15)



The form of the graph gives the idea that there might be asymptotes.

We start with the limits for φ tending to 0 and infinity.

```
(%i14)  limit([x(φ),y(φ)],φ,%pi);
```

(%o14) $\left[\lim_{\varphi\to\pi}\frac{\cos(\varphi)}{\frac{\sin(\varphi)}{d}-\frac{A\cos(\varphi)}{2}-\frac{A}{2}},d\right]$

```
(%i15)  limit(x(φ),φ,%pi);
Is d positive, negative or zero?p;
  (%o15)  infinity
(%i16)  limit([x(φ),y(φ)],φ,0);
```

(%o16) $\left[-\frac{1}{A},0\right]$

We can conclude that $y = d$ which is $y = 0.05$ is the horizontal asymptote. Next page shows additional explanation. The result given in `%o16` will become clear later.

Now for the second asymptote: We try to find its slope first. According to the sketch it is $\tan(\alpha)$ when $r(\alpha)$ tends to infinity which is equal to $\rho(\alpha)$ becomes 0.

So we have to solve the equation:

```
(%i18)  ρ(φ);
```
(%o18) $$\frac{\sin(\varphi)}{d}-\frac{A\cos(\varphi)}{2}-\frac{A}{2}$$
```
(%i19)  eq1:ρ(α)=0;
```
(eq1) $$\frac{\sin(\alpha)}{d}-\frac{A\cos(\alpha)}{2}-\frac{A}{2}=0$$

Unfortunately we cannot solve this equation in one step with Maxima (with *DERIVE* it is possible). There are some ways to overcome this deficiency. We will show tow of them:

```
(%i20)  solve(trigrat(eq1),α);
```
(%o20) $$[\sin(\alpha)=\frac{d\,A\cos(\alpha)+d\,A}{2}]$$
```
(%i21)  (2*sqrt(1-(cos(α)^2))=(d*A*cos(α)+d*A))^2;
```
(%o21) $$4\left(1-\cos(\alpha)^2\right)=(d\,A\cos(\alpha)+d\,A)^2$$
```
(%i22)  solve(%,α);
solve: using arc-trig functions to get a solution.
 Some solutions will be lost.
```
(%o22) $$[\alpha=\pi,\alpha=\pi-\operatorname{acos}\left(\frac{d^2A^2}{d^2A^2+4}-\frac{4}{d^2A^2+4}\right)]$$
```
(%i23)  tan(%pi-acos((d^2*A^2)/(d^2*A^2+4)-4/(d^2*A^2+4)));
```
(%o23) $$-\frac{\sqrt{1-\left(\frac{d^2A^2}{d^2A^2+4}-\frac{4}{d^2A^2+4}\right)^2}}{\frac{d^2A^2}{d^2A^2+4}-\frac{4}{d^2A^2+4}}$$
```
(%i24)  radcan(%);
```
(%o24) $$-\frac{4\,|d|\,|A|}{d^2A^2-4}$$
```
(%i25)  subst([d=1/20,A=20],%);
```
(%o25) $$\frac{4}{3}$$
```
(%i26)  tan(acos(3/5));
```
(%o26) $$\frac{4}{3}$$

We know that our asymptote has the form $y=\frac{4}{3}x+b$. We need to know b. But let me show a second way to find the slope of the asymptote:

```
(%i27) eq1:ρ(α)=0$
(%i28) eq2:sin(α)^2+cos(α)^2=1$
(%i29) eq3:tan(α)=sin(α)/cos(α)$
(%i30) solve(eliminate([eq1,eq2,eq3],[sin(α),cos(α)]),tan(α));
```

(%o30) $[\tan(\alpha)=-\frac{4\,d\,A}{d^2\,A^2-4},\tan(\alpha)=0]$

```
(%i31) subst([A=20,d=1/20],%);
```

(%o31) $[\tan(\alpha)=\frac{4}{3},\tan(\alpha)=0]$

The idea is to find the intersection point(s) of the asymptote and the graph in infinity. Is it possible to derive the implicit form of the graph? Let's try.

```
(%i32) eq1:Y=4/3*X+b$
(%i33) trigreduce(Y/X=y(φ)/x(φ));
```

(%o33) $\frac{Y}{X}=\tan(\varphi)$

```
(%i34) solve(Y/X=tan(φ),φ);
solve: using arc-trig functions to get a solution.
 Some solutions will be lost.
```

(%o34) $[\varphi=\operatorname{atan}\left(\frac{Y}{X}\right)]$

```
(%i36) t:atan(Y/X)$
       eq2:X=r(t)*cos(t);
```

(eq2) $X=\frac{1}{\sqrt{\frac{Y^2}{X^2}+1}\left(\frac{Y}{d\,X\sqrt{\frac{Y^2}{X^2}+1}}-\frac{A}{2\sqrt{\frac{Y^2}{X^2}+1}}-\frac{A}{2}\right)}$

Some manipulation is necessary to obtain a reasonable form of `eq2` (implicit form of the trajectory).

```
(%i37) eq2:radcan(eq2);
```

(eq2) $X=-\frac{2\,d\,X\,|X|}{d\,A\,X\sqrt{Y^2+X^2}-2\,|X|\,Y+d\,A\,X\,|X|}$

```
(%i38) eq2:eq2/X;
```

(eq2) $1=-\frac{2\,d\,|X|}{d\,A\,X\sqrt{Y^2+X^2}-2\,|X|\,Y+d\,A\,X\,|X|}$

```
(%i39) eq2:d*A*X*sqrt(Y^2+X^2)-2*abs(X)*Y+d*A*X*abs(X)=-2*d*abs(X);
```

(eq2) $d\,A\,X\sqrt{Y^2+X^2}-2\,|X|\,Y+d\,A\,X\,|X|=-2\,d\,|X|$

```
(%i40) eq2:radcan(eq2/X);
```

(eq2) $\frac{d\,A\,X\sqrt{Y^2+X^2}-2\,|X|\,Y+d\,A\,X\,|X|}{X}=-\frac{2\,d\,|X|}{X}$

```
(%i41) eq2:(d*A*sqrt(Y^2+X^2)=(-2*d+2*Y-d*A*X))^2;
```

(eq2) $d^2\,A^2\,(Y^2+X^2)=(2\,Y-d\,A\,X-2\,d)^2$

Looks good, seems to be a conic (hyperbola?).

```
(%i42) eq2:subst([A=20,d=5/100],eq2);
```

(eq2) $Y^2+X^2=\left(2\,Y-X-\frac{1}{10}\right)^2$

```
(%i43) solve([eq1,eq2],[Y,X]);
```

(%o43) $[\,[\,Y=\frac{15\,b-1}{300\,b-25}\,,\,X=-\frac{900\,b^2-120\,b+3}{1200\,b-100}\,]\,]$

```
(%i44) solve(300*b-25=0,b);
```

(%o44) $[\,b=\frac{1}{12}\,]$

```
(%i45) Y(X):=4/3*X+1/12$
```

We prepare the expressions for the plot and then plot the trajectory again but now together with the two asymptotes.

```
(%i46) as1:explicit(Y(X),X,-0.5,1)$
(%i47) as2:explicit(1/20, X,-2,0)$
(%i48) g1: subst([d=1/20,A=20],parametric(x(φ),y(φ), φ,0,%pi))$
(%i49) text:label(["trajectory for A=20,d=0.05",-0.4,0.15])$
-->    wxdraw2d(nticks=200,color=red,
           xaxis=true,yaxis=true,
           xrange=[-1,0.5],yrange=[0,0.2],
           title="y(x)",g1,color=blue,as1,color=blue,as2,
           color=black,text)$
```

(%t50)



We can do without complicated search for the asymptotes when we identify the graph as a hyperbola first. Then it is easy work to find its asymptotes.

As the first ideas for this collection of physics problems treated by means of computer algebra were realized with *DERIVE* we will finish with the *DERIVE* solution. We will also do the initial calculations to get the differential equation and we will find the asymptotes directly.

#1: [CaseMode := Sensitive, InputMode := Word]

#2: $$\left[\rho(\phi) :=, \; r(\phi) := \frac{1}{\rho(\phi)}\right]$$

#3: $$\frac{d}{d\phi}\, r(\phi) = - \frac{\rho'(\phi)}{\rho(\phi)^2}$$

#4: $$\frac{m}{2}\cdot\left(\frac{C^2}{r(\phi)^4}\cdot\left(\frac{d}{d\phi}\, r(\phi)\right)^2 + \frac{C^2}{r(\phi)^2}\right) + \frac{2\cdot Z\cdot e^2}{r(\phi)} = E_n$$

#5: $$\frac{C^2\cdot m\cdot \rho'(\phi)^2 + C^2\cdot m\cdot \rho(\phi)^2 + 4\cdot Z\cdot e^2\cdot \rho(\phi)}{2} = E_n$$

#6: $$\frac{\left(\dfrac{C^2\cdot m\cdot \rho'(\phi)^2 + C^2\cdot m\cdot \rho(\phi)^2 + 4\cdot Z\cdot e^2\cdot \rho(\phi)}{2} = E_n\right)\cdot 2}{C^2\cdot m}$$

#7: $$\frac{C^2\cdot m\cdot \rho'(\phi)^2 + C^2\cdot m\cdot \rho(\phi)^2 + 4\cdot Z\cdot e^2\cdot \rho(\phi)}{C^2\cdot m} = \frac{2\cdot E_n}{C^2\cdot m}$$

#8: $$\rho'(\phi)^2 + \rho(\phi)^2 + \frac{4\cdot Z\cdot e^2\cdot \rho(\phi)}{C^2\cdot m} = \frac{2\cdot E_n}{C^2\cdot m}$$

We reduce the differential equation:

#9: $$\rho'(\phi)^2 + \rho(\phi)^2 + A\cdot \rho(\phi) = B$$

#10: $$\frac{d}{d\phi}\left(\rho'(\phi)^2 + \rho(\phi)^2 + A\cdot \rho(\phi) = B\right)$$

#11: $$2\cdot \rho'(\phi)\cdot \rho''(\phi) + \rho'(\phi)\cdot (2\cdot \rho(\phi) + A) = 0$$

#12: $$\rho'(\phi)\cdot (2\cdot \rho''(\phi) + 2\cdot \rho(\phi) + A) = 0$$

#13: $$2\cdot \rho''(\phi) + 2\cdot \rho(\phi) + A = 0$$

#14: $$\rho''(\phi) + \rho(\phi) + \frac{A}{2} = 0$$

#15: $$\mathrm{DSOLVE2}\left(0,\; 1,\; -\frac{A}{2},\; \phi\right) = c1\cdot \mathrm{COS}(\phi) + c2\cdot \mathrm{SIN}(\phi) - 10$$

#16: $$\rho(\phi) := c1\cdot \mathrm{COS}(\phi) + c2\cdot \mathrm{SIN}(\phi) - \frac{A}{2}$$

Now we have solved the differential equation. For the syntax of DSOLVE2 see page 17. We find the constants in the same way as we did with Maxima.

$$\#16:\quad \rho(\phi) := c1\cdot \mathrm{COS}(\phi) + c2\cdot \mathrm{SIN}(\phi) - \frac{A}{2}$$

$$\#17:\quad \rho(\pi) = 0 = \left(- \frac{A}{2} - c1 = 0\right)$$

$$\#18:\quad c1 := - \frac{A}{2}$$

$$\#19:\quad \lim_{\phi\to\pi} r(\phi)\cdot \mathrm{SIN}(\phi) = d = \left(\frac{1}{c2} = d\right)$$

$$\#20:\quad c2 := \frac{1}{d}$$

We plot the trajectory with different ranges for φ:

$$\#21:\quad [A := 20,\ d := 0.05]$$

$$\#22:\quad [r(\phi)\cdot \mathrm{COS}(\phi),\ r(\phi)\cdot \mathrm{SIN}(\phi)]$$

$$\#23:\quad \left[- \frac{\mathrm{COS}(\phi)}{10\cdot(\mathrm{COS}(\phi) - 2\cdot \mathrm{SIN}(\phi) + 1)},\ - \frac{\mathrm{SIN}(\phi)}{10\cdot(\mathrm{COS}(\phi) - 2\cdot \mathrm{SIN}(\phi) + 1)}\right]$$



Now we can see the meaning of the Maxima result in %o16 above (right plot).

We find the implicit form of the curve and use a tool from the *DERIVE* Newsletter #83:

$$\#37:\quad 100\cdot(x^2 + y^2) = (10\cdot x - 20\cdot y + 1)^2$$

$$\#38:\quad \mathrm{ch\_conic}(100\cdot(x^2 + y^2) = (10\cdot x - 20\cdot y + 1)^2)$$

$$\#39:\quad \begin{bmatrix} \text{Type} & \text{Center} & \text{Real axis 2a} & \text{Imaginary axis 2b} \\ \text{Hyperbola} & \left[-\frac{1}{40}, \frac{1}{20}\right] & \frac{1}{20} & \frac{1}{10} \end{bmatrix}$$

$$\begin{bmatrix} \text{Real Vertices} & \text{Asymptotes} \\ \begin{bmatrix} -\frac{\sqrt{5}}{200} - \frac{1}{40} & \frac{\sqrt{5}}{100} + \frac{1}{20} \\ \frac{\sqrt{5}}{200} - \frac{1}{40} & \frac{1}{20} - \frac{\sqrt{5}}{100} \end{bmatrix} & \begin{bmatrix} \frac{3\cdot\sqrt{5}\cdot t}{10} - \frac{1}{40} & \frac{2\cdot\sqrt{5}\cdot t}{5} + \frac{1}{20} \\ \frac{\sqrt{5}\cdot t}{2} - \frac{1}{40} & \frac{1}{20} \end{bmatrix} \end{bmatrix}$$

$$\#40: \quad \text{SOLVE}\left(x = \frac{3\cdot\sqrt{5}\cdot t}{10} - \frac{1}{40} \wedge y = \frac{2\cdot\sqrt{5}\cdot t}{5} + \frac{1}{20}, [y, t]\right)$$

$$\#41: \quad y = \frac{16\cdot x + 1}{12} \wedge t = \frac{2\cdot\sqrt{5}\cdot x}{3} + \frac{\sqrt{5}}{60}$$

Additional Explanation

The total energy and angular momentum of the particle are constant.

Total energy of the particle is the sum of kinetic energy and potential energy

$$E_k + E_p = E_n = const$$

Kinetic energy:

$$E_k = \frac{1}{2} m v^2 = \frac{1}{2} m (v_x{}^2 + v_y{}^2) = \frac{1}{2} m (v_r{}^2 + v_\varphi{}^2)$$

v_r and v_φ are the polar components of the velocity: $v_r = \frac{dr}{dt} = \dot{r},\ v_\varphi = r\frac{d\varphi}{dt} = r\dot{\varphi}.$

So we have in polar coordinates:

$$E_k = \frac{1}{2} m (v_r{}^2 + v_\varphi{}^2) = \frac{1}{2} m (\dot{r}^2 + r^2\dot{\varphi}^2).$$

Potential energy:

($2e$ = charge of the α-particle, $Z\,e$ = charge of the nucleus with Z protons)

$$E_p = \frac{Q_1 Q_2}{r} = \frac{Q_\alpha Q_z r}{r} = \frac{2e\,Ze}{r} = \frac{2Ze^2}{r}$$

Magnitude of the angular momentum:

$$L = |\vec{r} \times m\vec{v}| = |r\vec{e}_r \times m(\dot{r}\vec{e}_r + r\dot{\varphi}\vec{e}_r)| = |m r^2 \dot{\varphi}| = const \ \rightarrow\ r^2\dot{\varphi} = C$$

$(\vec{e}_r \times \vec{e}_r = 1,\ \vec{e}_r \times \vec{e}_\varphi = 0)$